
UNIQUE 

Returns a list of unique values from a range.  

The worksheet below contains a list or orders for a company that sells ice cream. The owner 
of the business needs a list of customers to be displayed in column F. This could be done by 
copying the list from column B into column F and then removing the duplicates by using the 
Data > Remove Duplicates command on the Ribbon. 

However, the Unique function offers a much simpler way to obtain the same information. 

Cell F2 contains the following: =UNIQUE(B2:B19) 

This tells Excel to look at the range B2:B19 and list one instance of each value. 

 

Unlike traditional formulas and functions, it is not necessary to manually copy the function 
down column F. The cells below F2 are automatically populated. Each cell contains the same 
function =UNIQUE (B2:B19) 

Filling multiple cells with a single formula or function is called spilling and the populated 
range of cells is called the spill range. 

A blue border appears around F2:F10, indicating that the function in F2 is a dynamic array 
function. The formulas in F3:F10 cannot be edited – they appear dimmed out in the formula 
bar. Any amendments must be made in F2 



theexceltrainer.co.uk   2 

If one of the cells in the range B2:B19 is changed and the value entered is a new value (i.e. it 
does not already exist in B2:B19), the spill range automatically expands, so in this example 
the new value will appear in F11 

#SPILL 

In the screenshot below, there is data in F6. The dynamic array function in F2 should return 
9 values (the blue border is around F2:F10).  

Rather than overwriting the contents of F6 with a value from the dynamic array, Excel 
displays #SPILL in F2. As soon as the current contents of F6 are deleted, the values from the 
dynamic array function “spill” down column F. 

 

Dynamic Arrays and The # Sign 

Typing a hash sign (#) within a formula that reference a dynamic array function will 
reference the entire spill range. In the example below: 

• The formula in I1 is =COUNTA(F2) 

• The formula in I2 is =COUNTA(F2#) 

• The formula in I3 is =COUNTA(F2:F10) 

The difference between the formulas in I2 and I3 is that the formula in I2 refers to the spill 
range that starts in F2 (which may grow or shrink), whereas the formula in I3 refers to a 
fixed range F2:F10. 

UNIQUE Function: By_Col Parameter 

By default, the Unique function looks down a column and extracts unique values from the 
rows within that column. To force Excel to look across a row and extract unique values from 
the columns within that row, set the second (optional) parameter of the function to TRUE 

 

In the spreadsheet above, the formula in A8 is =UNIQUE(B2:G2,TRUE) 

Notice that when using the TRUE parameter, the results spill across a row. To force the spill 
range to be a column, wrap the above formula in the TRANSPOSE function… 

=TRANSPOSE(UNIQUE(B2:G2,TRUE)) 

UNIQUE Function: Exactly_Once Parameter 

By default, the Unique function displays a list of all unique values regardless of how many 
times they appear.  

But what if what was needed was a list of customers who made a single purchase? To do 
that, set the 3rd (optional) parameter to TRUE. 

The formula in the example below is: =UNIQUE(B2:B19,FALSE,TRUE) 

Note that the second parameter has been set to FALSE to indicate that Excel needs to look 
down a column and extracts unique values from the rows within that column. 

SORT 

Sorts the value in a range into ascending or descending order.  

The worksheet below contains a list or orders for a company that sells ice cream. The owner 
of the business needs an alphabetical list of customers.  

One way to do this is to use the Sort command on the Ribbon to sort A2:A19 into ascending 
order. However, the owner wants to keep this list “as is”. Another option is to copy the list 
to column D and use the Sort command on the Ribbon to sort that list. But that would 
increase the size of the file as well as making more work. 

The Sort function offers a much simpler way to do the same thing. 

Cell D2 contains the following: =SORT(B2:B19) 

SORT Function: Sort_Index Parameter 

If the range to be sorted contains more than 1 column & needs to be sorted by a column 
other than the first one, use (optional) Sort_Index to indicate which column to sort by.  

In the example below, the owner of the ice cream business needs, in columns F:I, a copy of 
the table in A2:D2 sorted by Revenue. The formula in F2 is: =SORT(A2:D19,4) 

This tells Excel to sort the range A4:D19 by the 4th column (the one containing the revenue) 

 

 

SORT Function: Sort_Order Parameter 

By default, the Sort function sorts in ascending order. To sort in descending order (highest 
to lowest, Z-A), set the 3rd (optional) parameter to -1. 

For example: =SORT(A2:D19,4,-1) will produce a copy of the data in A2:D19, sorted by 
revenue with the highest revenue listed first:  

 

 

SORT Function: By_Col Parameter 

By default, the SORT function will sort data vertically by rows. To sort a range horizontally by 
columns, set the fourth (optional) argument, by_col, to TRUE. 

SORTBY 

Allows you to sort a range based on a column or row that doesn’t appear in the results 

In the previous example, the requirement was to sort the list in columns A:D based on the 
revenue (column D) 

 

 

Using a different scenario, here I have a list of the sales reps at the ice cream company 
together with the revenue that each has generated. The owner of the business wants a list 
of the sales reps based on their revenue but does not want the revenue to be displayed. 

The formula in D2 is : =SORTBY(A2:A6,B2:B6,-1).  

In other words, sort the range A2:A6 based on the corresponding values in B2:B6 

 

SORTBY Function: Sort_Order Parameter 

The 3rd (optional) parameter defines the sort order. Use 1 (or do not specify) for ascending 
or -1 for descending. In this example, -1 has been specified so the names are listed based on 
the revenue generated in highest-lowest order (Chris being the highest revenue generator) 

Combining UNIQUE and SORT 

Requirement: A unique list of customers sorted alphabetically by name. To do this, combine 
the UNIQUE and SORT functions.  

In the example below, D2 contains: =SORT(UNIQUE(A2:A18)) 

 

  



theexceltrainer.co.uk   9 

FILTER 

Extracts a set of matching records from a dataset based on supplied criteria.  

In the example below, starting at F4 are the records columns A:D that match the flavour 
shown in G1. 

The formula in F4 is =FILTER(A2:D19,A2:A19=G1) 

The 1st parameter is the range containing the data to be filtered 

The 2nd parameter is the criteria. In this example this means “only display the records 
where whatever is displayed in G1 appears in A2:A19” 

 

 

Filter Function: Is_Empty Parameter 

If there are no records that match the criteria, the FILTER function returns a #CALC! error. 
To avoid this, add some text, inside double quotes, which will be displayed instead.  

In the example below, B4 contains =FILTER(A2:D19,A2:A19=G1,"No Matches Found") 

